

# SPARQL

Mauricio Espinoza Mejía

[mauricio.espinoza@ucuenca.edu.ec](mailto:mauricio.espinoza@ucuenca.edu.ec)

# Content

- Main Idea of SPARQL Queries
- Components of SPARQL Queries
- Graph Patterns
- Constraints on Solutions
- Components of SPARQL Queries: Other features
- Solution Modifiers
- SPARQL 1.1


# SPARQL in General

## A family of W3C recommendations

- **SPARQL Query**
  - Declarative query language for RDF data
- **SPARQL Update**
  - Declarative update language for RDF data
- **SPARQL Protocol**
  - Communication between SPARQL processing services (a.k.a. SPARQL endpoints) and clients
- **SPARQL Query Results XML Format**
  - XML format for serializing query results


# Main Idea of SPARQL Queries

- **Pattern matching:**
  - Describe subgraphs of the queried RDF graph
  - Subgraphs that match your description contribute an answer
  - Mean: graph patterns (i.e. RDF graphs with variables)




# Main Idea of SPARQL Queries

Queried RDF graph:



Result:

| <b>?v</b>           |
|---------------------|
| dbpedia:Mount_Baker |
| dbpedia:Mount_Etna  |



# Content

- Main Idea of SPARQL Queries
- Components of SPARQL Queries
- Graph Patterns
- Constraints on Solutions
- Components of SPARQL Queries: Other features
- Solution Modifiers
- SPARQL 1.1

# Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

# Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Prologue:**

- Prefix definitions for using compact URIs
- Attention (difference to Turtle): No period (".") character as separator



# Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Result form specification:**
  - SELECT, DESCRIBE, CONSTRUCT, or ASK

# Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Dataset specification:**
  - Specify the RDF dataset to be queried

# Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Query Pattern:**
  - WHERE clause specifies the graph pattern to be matched

# Content

- Main Idea of SPARQL Queries
- Components of SPARQL Queries
- Graph Patterns
- Constraints on Solutions
- Components of SPARQL Queries: Other features
- Solution Modifiers
- SPARQL 1.1

# Graph Patterns

- **Different types of graph patterns for the query pattern**
- **(WHERE clause):**
  - Basic graph pattern (BGP)
  - Group graph pattern
  - Optional graph pattern
  - Union graph pattern
  - Graph graph pattern
  - (Constraints)

# Basic Graph Patterns

- Set of triple patterns (i.e. RDF triples with variables)
- Variable names prefixed with “?” or “\$” (e.g. ?v, \$v)
- Turtle syntax

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?name
WHERE {
    ?v rdf:type umbel-sc:Volcano .
    ?v rdfs:label ?name .
}
```

# Basic Graph Patterns

- Set of triple patterns (i.e. RDF triples with variables)
- Variable names prefixed with “?” or “\$” (e.g. ?v, \$v)
- Turtle syntax

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?name
WHERE {
    ?v rdf:type umbel-sc:Volcano ;
        rdfs:label ?name .
}
```

# Basic Graph Patterns (Example)

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;           Data*
                    rdfs:label "Etna" .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano,
                    umbel-sc:NaturalElevation ;
                    rdfs:label "Beerenberg"@en ;
                    rdfs:label "Беренберг"@ru .
```

- Question: What are the names of all (known) volcanos?

```
SELECT ?name WHERE {                                     Query*
  ?v rdf:type umbel-sc:Volcano ;
  rdfs:label ?name . }
```

Result:

| ?name           |
|-----------------|
| "Etna"          |
| "Беренберг"@ru  |
| "Beerenberg"@en |




# Basic Graph Patterns (Example)

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;           Data*
                    rdfs:label "Etna" .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano,
                    umbel-sc:NaturalElevation ;
                    rdfs:label "Beerenberg"@en ;
                    rdfs:label "Бееренберг"@ru  .
```

Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {                                     Query
  ?v rdf:type ?type ;
    rdfs:label "Beerenberg" .
}
```

| ?type |
|-------|
|       |



# Basic Graph Patterns (Example)

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;           Data*
                    rdfs:label "Etna" .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano,
                        umbel-sc:NaturalElevation ;
                    rdfs:label "Beerenberg"@en ;
                    rdfs:label "Бееренберг"@ru .
```

Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {                                     Query
  ?v rdf:type ?type ;
    rdfs:label "Beerenberg"@en .
}
```

| ?type                     |
|---------------------------|
| umbel-sc:Volcano          |
| umbel-sc:NaturalElevation |

# Basic Graph Patterns (Example)

```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;      Data
                    p:location dbpedia:United_States .
dbpedia:United_States rdfs:label "United States" .
```

- **Question: Where are all (known) volcanos located?**  
**(List the names of these locations)**
- **Blank nodes in SPARQL queries**
  - As subject or object of a triple pattern

```
SELECT ?name WHERE {                                Query
  _:x rdf:type umbel-sc:Volcano ;
      p:location [ rdfs:label ?name ] . }
```

| ?name           |
|-----------------|
| "United States" |

# Basic Graph Patterns (Example)

```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;      Data
    p:location [ rdfs:label "United States"@en ,
                  "États-Unis"@fr ] .
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;
    p:location [ rdfs:label "Italy" ] .
```

- **Blank nodes in the queried data**
  - Blank node identifiers may occur in the results

```
SELECT ?l ?name WHERE {      Query
    ?v rdf:type umbel-sc:Volcano ;
    p:location ?l .
    ?l rdfs:label ?name .
}
```

| ?l  | ?name              |
|-----|--------------------|
| _:x | "United States"@en |
| _:x | "États-Unis"@fr    |
| _:y | "Italy"            |

# Optional Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en .
```

Data

- Question: What are all (known) volcanos and their names?

```
SELECT ?v ?name WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
  rdfs:label ?name . }
```

Query

- Problem: Mount Baker **missing** (it has no name)

| ?v                 | ?name           |
|--------------------|-----------------|
| dbpedia:Mount_Etna | "Etna"          |
| dbpedia:Beerenberg | "Beerenberg"@en |

# Optional Graph Patterns

- Keyword **OPTIONAL** allows for optional patterns

```
SELECT ?v ?name WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  OPTIONAL { ?v rdfs:label ?name }  
}
```

Query

| ?v                  | ?name           |
|---------------------|-----------------|
| dbpedia:Mount_Etna  | "Etna"          |
| dbpedia:Mount_Baker |                 |
| dbpedia:Beerenberg  | "Beerenberg"@en |

- Optional patterns may result in unbound variables

# Union Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;           Data
    rdfs:label "Etna" ;
    p:location dbpedia:Italy .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;
    p:location dbpedia:United_States .
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;
    rdfs:label "Beerenberg"@en ;
    p:location dbpedia:Norway .
```

- Question: What volcanos are located in Italy or in Norway?

```
SELECT ?v WHERE {                                       Query
    ?v rdf:type umbel-sc:Volcano ;
    p:location    ?    . }
```

# Union Graph Patterns

- Union graph patterns allow for alternatives

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
                  { ?v p:location dbpedia:Italy }  
                  UNION  
                  { ?v p:location dbpedia:Norway }  
}
```

Semantically  
equivalent

```
SELECT ?v WHERE {  
  { ?v rdf:type umbel-sc:Volcano ;  
    p:location dbpedia:Italy }  
  UNION  
  { ?v rdf:type umbel-sc:Volcano ;  
    p:location dbpedia:Norway }  
}
```



# Group Graph Patterns

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
                  { ?v p:location dbpedia:Italy }  
                  UNION  
                  { ?v p:location dbpedia:Norway }  
}
```

Query

Semantically equivalent

```
SELECT ?v WHERE { { ?v rdf:type umbel-sc:Volcano }  
                  { { ?v p:location dbpedia:Italy }  
                    UNION  
                    { ?v p:location dbpedia:Norway } }  
}
```

Query

Practica